
  Installation



  
    
  
  Using Igniter


The easiest way to get started is to use the Igniter installer.
mix archive.install hex igniter_new
mix igniter.install inertia

The following options can be used to customize the installation:
--client-framework [react|vue|svelte] # Configures the client-side framework in `assets/package.json`
--camelize-props                      # Sets `camelize_props: true` in `config.exs` (See below)
--history-encrypt                     # Sets `history: [encrypt: true]` in `config.exs` (See below)
--typescript                          # Creates a TypeScript config file and installs dev dependencies


  
    
  
  Manually


The package can be installed by adding inertia to your list of dependencies in mix.exs:
def deps do
  [
    {:inertia, "~> 2.5.1"}
  ]
end
Add your desired configuration in your config.exs file:
# config/config.exs

config :inertia,
  # The Phoenix Endpoint module for your application. This is used for building
  # asset URLs to compute a unique version hash to track when something has
  # changed (and a reload is required on the frontend).
  endpoint: MyAppWeb.Endpoint,

  # An optional list of static file paths to track for changes. You'll generally
  # want to include any JavaScript assets that may require a page refresh when
  # modified.
  static_paths: ["/assets/app.js"],

  # The default version string to use (if you decide not to track any static
  # assets using the `static_paths` config). Defaults to "1".
  default_version: "1",

  # Enable automatic conversion of prop keys from snake case (e.g. `inserted_at`),
  # which is conventional in Elixir, to camel case (e.g. `insertedAt`), which is
  # conventional in JavaScript. Defaults to `false`.
  camelize_props: false,

  # Instruct the client side whether to encrypt the page object in the window history
  # state. This can also be set/overridden on a per-request basis, using the `encrypt_history`
  # controller helper. Defaults to `false`.
  history: [encrypt: false],

  # Enable server-side rendering for page responses (requires some additional setup,
  # see instructions below). Defaults to `false`.
  ssr: false,

  # Whether to raise an exception when server-side rendering fails (only applies
  # when SSR is enabled). Defaults to `true`.
  #
  # Recommended: enable in non-production environments and disable in production,
  # so that SSR failures will not cause 500 errors (but instead will fallback to
  # CSR).
  raise_on_ssr_failure: config_env() != :prod
This library includes a few modules to help render Inertia responses:
	Inertia.Plug: a plug for detecting Inertia.js requests and preparing the connection accordingly.
	Inertia.Controller: controller functions for rendering Inertia.js-compatible responses.
	Inertia.HTML: HTML components for Inertia-powered views.

To get started, import Inertia.Controller in your controller helper and Inertia.HTML in your html helper:
  # lib/my_app_web.ex
  defmodule MyAppWeb do
    def controller do
      quote do
        use Phoenix.Controller, namespace: MyAppWeb

+       import Inertia.Controller
      end
    end

    def html do
      quote do
        use Phoenix.Component

+       import Inertia.HTML
      end
    end
  end
Then, install the plug in your browser pipeline:
  # lib/my_app_web/router.ex
  defmodule MyAppWeb.Router do
    use MyAppWeb, :router

    pipeline :browser do
      plug :accepts, ["html"]

+     plug Inertia.Plug
    end
  end
Next, replace the title tag in your layout with the <.inertia_title> component, so that the client-side library will keep the title in sync, and add the <.inertia_head> component:
  # lib/my_app_web/components/layouts/root.html.heex
  <!DOCTYPE html>
  <html lang="en" class="[scrollbar-gutter:stable]">
    <head>
-     <.live_title>{@page_title}</.live_title>
+     <.inertia_title>{@page_title}</.inertia_title>
+     <.inertia_head content={@inertia_head} />
    </head>
You're now ready to start rendering inertia responses!

  
    
  
  Rendering responses


Rendering an Inertia.js response looks like this:
defmodule MyAppWeb.ProfileController do
  use MyAppWeb, :controller

  def index(conn, _params) do
    conn
    |> assign_prop(:text, "Hello world")
    |> render_inertia("ProfilePage")
  end
end
The assign_prop function allows you to define props that should be passed in to the component. The render_inertia function accepts the conn, the name of the component to render, and an optional map containing more initial props to pass to the page component.
This action will render an HTML page containing a <div> element with the name of the component and the initial props, following Inertia.js conventions. On subsequent requests dispatched by the Inertia.js client library, this action will return a JSON response with the data necessary for rendering the page.
If you want to automatically convert your prop keys from snake case (conventional in Elixir) to camel case to keep with JavaScript conventions (e.g. first_name to firstName), you can configure that globally or enable/disable it on a per-request basis.
import Config

config :inertia,
  endpoint: MyAppWeb.Endpoint,
  camelize_props: true
defmodule MyAppWeb.ProfileController do
  use MyAppWeb, :controller

  def index(conn, _params) do
    conn
    |> assign_prop(:first_name, "Bob")
    |> camelize_props()
    |> render_inertia("ProfilePage")
  end
end

  
    
  
  Setting up the client-side


The Inertia.js docs provide a good general walk-through on how to setup your JavaScript assets to boot your Inertia app. If you're new to Inertia, we recommend checking that out to familiarize yourself with how it all works. Here we'll provide some guidance on getting your Phoenix app with esbuild configured for basic client-side rendering (and further down, we'll delve into server-side rendering).
To get started, install the Inertia.js library for the frontend framework of your choice. In these instructions we'll use React, but the process is similar for other Inertia-compatible frameworks, like Vue or Svelte.
cd assets
npm install @inertiajs/react react react-dom
Replace the contents of your app.js file with the Inertia boot function and rename it to app.jsx (since we are using JSX).
// assets/js/app.jsx

import React from "react";
import axios from "axios";

import { createInertiaApp } from "@inertiajs/react";
import { createRoot } from "react-dom/client";

axios.defaults.xsrfHeaderName = "x-csrf-token";

createInertiaApp({
  resolve: async (name) => {
    return await import(`./pages/${name}.jsx`);
  },
  setup({ App, el, props }) {
    createRoot(el).render(<App {...props} />);
  },
});
The example above assumes your pages live in the assets/js/pages directory and have a default export with page component, like this:
// assets/js/pages/Dashboard.jsx

import React from "react";

const Dashboard = () => {
  return (
    <div>
      {/* ... page contents ...*/}
    </div>
  );
}

export default Dashboard;
Next, make some adjustments to your esbuild config:
	Ensure the version is >= 0.19.0 (this is required for glob-style imports for your pages)
	Update your entrypoint filename to the correct .jsx extension
	Ensure your build --target is at least es2020

# config/config.exs

config :esbuild,
  version: "0.21.5",
  my_app: [
    args: ~w(js/app.jsx --bundle --target=es2020 --outdir=../priv/static/assets --external:/fonts/* --external:/images/*),
    cd: Path.expand("../assets", __DIR__),
    env: %{"NODE_PATH" => Path.expand("../deps", __DIR__)}
  ]
If you updated your esbuild version, you'll need to run mix esbuild.install to fetch the new version.
Esbuild also supports code-splitting, which can be useful for larger applications. To enable it, you'll need to:
	Set the format as esm
	Add the --splitting flag
	Optionally, set the chunk-names flag to customize the output filenames

  # config/config.exs

  config :esbuild,
    version: "0.21.5",
    my_app: [
-     args: ~w(js/app.jsx --bundle --target=es2020 --outdir=../priv/static/assets --external:/fonts/* --external:/images/*),
+     args: ~w(js/app.jsx --bundle --chunk-names=chunks/[name]-[hash] --splitting --format=esm --target=es2020 --outdir=../priv/static/assets --external:/fonts/* --external:/images/*),
      cd: Path.expand("../assets", __DIR__),
      env: %{"NODE_PATH" => Path.expand("../deps", __DIR__)}
    ]
After that, we need to update our root layout to load the JavaScript bundle as an ESM module (by changing the type attribute from text/javascript to module):
  # lib/my_app_web/components/layouts/root.html.eex

-  <script type='text/javascript' defer phx-track-static src={~p"/assets/app.js"}></script>
+  <script type='module' defer phx-track-static src={~p"/assets/app.js"}></script>
[!NOTE]
ESM code splitting requires modern browser support.
While most current browsers support ESM modules, you should verify compatibility requirements with your target audience.
You can read more about how code-splitting works with esbuild in the official documentation.


  
    
  
  Lazy data evaluation


If you have expensive data for your props that may not always be required (that is, if you plan to use partial reloads), you can wrap your expensive computation in a function and pass the function reference when setting your Inertia props. You may use either an anonymous function (or named function reference) and optionally wrap it with the Inertia.Controller.inertia_optional/1 function.
[!NOTE]
inertia_optional props will only be included the when explicitly requested in a partial
reload. If you want to include the prop on first visit, you'll want to use a
bare anonymous function or named function reference instead. See below for
examples of how prop assignment behaves.

Here are some specific examples of how the methods of lazy data evaluation differ:
conn
# ALWAYS included on first visit...
# OPTIONALLY included on partial reloads...
# ALWAYS evaluated...
|> assign_prop(:cheap_thing, cheap_thing())

# ALWAYS included on first visit...
# OPTIONALLY included on partial reloads...
# ONLY evaluated when needed...
|> assign_prop(:expensive_thing, fn -> calculate_thing() end)
|> assign_prop(:another_expensive_thing, &calculate_another_thing/0)

# NEVER included on first visit...
# OPTIONALLY included on partial reloads...
# ONLY evaluated when needed...
|> assign_prop(:super_expensive_thing, inertia_optional(fn -> calculate_thing() end))

  
    
  
  Deferred props


Requires Inertia v2.x on the client-side.
If you have expensive data that you'd like to automatically fetch (from the client-side via an async background request) after the page is initially rendered, you can mark the prop as deferred:
conn
|> assign_prop(:expensive_thing, inertia_defer(fn -> calculate_thing() end))
The inertia_defer/1 helper accepts a function argument in the first position. You may optionally use the inertia_defer/2 helper, which accepts a "group" name in the second position:
conn
|> assign_prop(:expensive_thing, inertia_defer(fn -> calculate_thing() end, "dashboard"))
If no group names are specified, then the client-side will issue a single async request to fetch all the deferred props. If there are multiple group names, then the client-side will issue one async request per group instead. This is useful if you have some very expensive data that you'd prefer fetch in parallel alongside other expensive data.

  
    
  
  Merge props


Requires Inertia v2.x on the client-side.
If you have prop data that should get merged with the existing data on the client-side on subsequent requests (for example, an array of paginated data being presented in an "infinite scroll" interface), then you can tag the prop value using the inertia_merge/1 helper:
conn
|> assign_prop(:paginated_list, inertia_merge(["a", "b", "c"]))
Merge props can also accept deferred props:
conn
|> assign_prop(:paginated_list, inertia_defer(&calculate_next_page/0) |> inertia_merge())
If you are working with complex data structures or nested objects you can use inertia_deep_merge(value)
conn
|> assign_prop(:complex_object, inertia_deep_merge(%{a: %{b: %{c: %{d: 1}}}}))

  
    
  
  Shared data


To share data on every request, you can use the assign_prop/2 function inside of a shared plug in your response pipeline. For example, suppose you have a UserAuth plug responsible for fetching the currently-logged in user and you want to be sure all your Inertia components receive that user data. Your plug might look something like this:
defmodule MyApp.UserAuth do
  import Inertia.Controller
  import Phoenix.Controller
  import Plug.Conn

  def authenticate_user(conn, _opts) do
    user = get_user_from_session(conn)

    # Here we are storing the user in the conn assigns (so
    # we can use it for things like checking permissions later on),
    # AND we are assigning a serialized represention of the user
    # to our Inertia props.
    conn
    |> assign(:user, user)
    |> assign_prop(:user, serialize_user(user))
  end

  # ...
end
Anywhere this plug is used, the serialized user prop will be passed to the Inertia component.

  
    
  
  Validations


Validation errors follow some specific conventions to make wiring up with Inertia's form helpers seamless. The errors prop is managed by this library and is always included in the props object for Inertia components. (When there are no errors, the errors prop will be an empty object).
The assign_errors function is how you tell Inertia what errors should be represented on the front-end. By default, you can either pass an Ecto.Changeset struct or a bare map to the assign_errors function. For other error data types, you may implement the Inertia.Errors protocol (see the Inertia.Errors module docs for more information).
def update(conn, params) do
  case MyApp.Settings.update(params) do
    {:ok, _settings} ->
      conn
      |> put_flash(:info, "Settings updated")
      |> redirect(to: ~p"/settings")

    {:error, changeset} ->
      conn
      |> assign_errors(changeset)
      |> redirect(to: ~p"/settings")
  end
end
The assign_errors function will automatically convert the changeset errors into a shape compatible with the client-side adapter. Since Inertia.js expects a flat map of key-value pairs, the error serializer will flatten nested errors down to compound keys:
{
  "name" => "can't be blank",

  // Nested errors keys are flattened with a dot separator (`.`)
  "team.name" => "must be at least 3 characters long",

  // Nested arrays are zero-based and indexed using bracket notation (`[0]`)
  "items[1].price" => "must be greater than 0"
}
Errors are automatically preserved across redirects, so you can safely respond with a redirect back to page where the form lives to display form errors.
If you need to construct your own map of errors (rather than pass in a changeset), be sure it's a flat mapping of atom (or string) keys to string values like this:
conn
|> assign_errors(%{
  name: "Name can't be blank",
  password: "Password must be at least 5 characters"
})

  
    
  
  Flash messages


This library automatically includes Phoenix flash data in Inertia props, under the flash key.
For example, given the following controller action:
def update(conn, params) do
  case MyApp.Settings.update(params) do
    {:ok, _settings} ->
      conn
      |> put_flash(:info, "Settings updated")
      |> redirect(to: ~p"/settings")

    {:error, changeset} ->
      conn
      |> assign_errors(changeset)
      |> redirect(to: ~p"/settings")
  end
end
When Inertia (or the browser) redirects to the /settings page, the Inertia component will receive the flash props:
{
  "component": "...",
  "props": {
    "flash": {
      "info": "Settings updated"
    },
    // ...
  }
}

  
    
  
  CSRF protection


This library automatically sets the XSRF-TOKEN cookie for use by the Axios client on the front-end. Since Phoenix expects to receive the CSRF token via the x-csrf-token header, you'll need to configure Axios in your front-end JavaScript to use that header name:
// assets/js/app.js

import axios from "axios";
axios.defaults.xsrfHeaderName = "x-csrf-token";

// the rest of your Inertia client code...

  
    
  
  History


Requires Inertia v2.x on the client-side.

  
    
  
  Encryption


If your page props contain sensitive data (such as information about the currently-authenticated user), you can opt to encrypt the history data that's cached in the browser.
conn
|> encrypt_history()
You can also enable history encryption globally in your application config:
config :inertia,
  history: [encrypt: true]

  
    
  
  Clearing history


To instruct the client to clear this history (for example, when a user logs out), you can use the clear_history/1 helper when building your response.
conn
|> clear_history()

  
    
  
  Testing


The Inertia.Testing module includes helpers for testing your Inertia controller responses, such as the inertia_component/1 and inertia_props/1 functions.
use MyAppWeb.ConnCase

import Inertia.Testing

describe "GET /" do
  test "renders the home page", %{conn: conn} do
    conn = get("/")
    assert inertia_component(conn) == "Home"
    assert %{user: %{id: 1}} = inertia_props(conn)
  end
end
use MyAppWeb.ConnCase

import Inertia.Testing

describe "POST /users" do
  test "fails when name empty", %{conn: conn} do
    conn = post("/users", %{"name" => ""})

    assert %{user: %{id: 1}} = inertia_props(conn)
    assert redirected_to(conn) == ~p"/users"
    assert inertia_errors(conn) == %{"name" => "can't be blank"}
  end
end
We recommend importing Inertia.Testing in your ConnCase helper, so that it will be at the ready for all your controller tests:
defmodule MyApp.ConnCase do
  use ExUnit.CaseTemplate

  using do
    quote do
      import Inertia.Testing

      # ...
    end
  end
end

  
    
  
  Server-side rendering


The Inertia.js client library comes with with server-side rendering (SSR) support, which means you can have your Inertia-powered client hydrate HTML that has been pre-rendered on the server (instead of performing the initial DOM rendering).
[!NOTE]
The steps for enabling SSR in Phoenix are similar to other backend frameworks, but instead of running a separate Node.js server process to render HTML, this library spins up a pool of Node.js process workers to handle SSR calls and manages the state of those node processes from your Elixir process tree. This is mostly just an implementation detail that you don't need to be concerned about, but we'll highlight how our ssr.js script differs from the Inertia.js docs.


  
    
  
  Add a server-side rendering module


You'll need to create a JavaScript module that exports a render function to perform the actual server-side rendering of pages. For the purpose of these instructions, we'll assume you're using React. The steps would be similar for other front-end environments supported by Inertia.js, such as Vue and Svelte.
Suppose your main app.jsx file looks something like this:
// assets/js/app.jsx

import React from "react";
import { createInertiaApp } from "@inertiajs/react";
import { createRoot } from "react-dom/client";

createInertiaApp({
  resolve: async (name) => {
    return await import(`./pages/${name}.jsx`);
  },
  setup({ App, el, props }) {
    createRoot(el).render(<App {...props} />);
  },
});
You'll need to create a second JavaScript file (alongside your app.jsx) that exports a render function. Let's name it ssr.jsx.
// assets/js/ssr.jsx

import React from "react";
import ReactDOMServer from "react-dom/server";
import { createInertiaApp } from "@inertiajs/react";

export function render(page) {
  return createInertiaApp({
    page,
    render: ReactDOMServer.renderToString,
    resolve: async (name) => {
      return await import(`./pages/${name}.jsx`);
    },
    setup: ({ App, props }) => <App {...props} />,
  });
}
This is similar to the server entry-point documented here, except we are simply exporting a function called render, instead of starting a Node.js server process.
Next, configure esbuild to compile the ssr.jsx bundle.
  # config/config.exs

  config :esbuild,
    version: "0.21.5",
    app: [
      args: ~w(js/app.jsx --bundle --target=es2020 --outdir=../priv/static/assets --external:/fonts/* --external:/images/*),
      cd: Path.expand("../assets", __DIR__),
      env: %{"NODE_PATH" => Path.expand("../deps", __DIR__)}
    ],
+   ssr: [
+     args: ~w(js/ssr.jsx --bundle --platform=node --outdir=../priv --format=cjs),
+     cd: Path.expand("../assets", __DIR__),
+     env: %{"NODE_PATH" => Path.expand("../deps", __DIR__)}
+   ]
Add the ssr build to the watchers in your dev environment, alongside the other asset watchers:
  # config/dev.exs
  config :my_app, MyAppWeb.Endpoint,
    # Binding to loopback ipv4 address prevents access from other machines.
    # Change to `ip: {0, 0, 0, 0}` to allow access from other machines.
    http: [ip: {127, 0, 0, 1}, port: 4000],
    check_origin: false,
    code_reloader: true,
    debug_errors: true,
    secret_key_base: "4Z2yyTu6Uy8AM+MguG3oldEf4aIdswR2BsCm1OtqDK0lEv++T02KktRaXfMbC/Zs",
    watchers: [
      esbuild: {Esbuild, :install_and_run, [:app, ~w(--sourcemap=inline --watch)]},
+     ssr: {Esbuild, :install_and_run, [:ssr, ~w(--sourcemap=inline --watch)]},
      tailwind: {Tailwind, :install_and_run, [:my_app, ~w(--watch)]}
    ]
Add the ssr build step to the asset build and deploy scripts.
  # mix.exs

  defp aliases do
    [
      setup: ["deps.get", "ecto.setup", "assets.setup", "assets.build"],
      "ecto.setup": ["ecto.create", "ecto.migrate", "run priv/repo/seeds.exs"],
      "ecto.reset": ["ecto.drop", "ecto.setup"],
      test: ["ecto.create --quiet", "ecto.migrate --quiet", "test"],
      "assets.setup": ["tailwind.install --if-missing", "esbuild.install --if-missing"],
-     "assets.build": ["tailwind app", "esbuild app"],
+     "assets.build": ["tailwind app", "esbuild app", "esbuild ssr"],
      "assets.deploy": [
        "tailwind app --minify",
        "esbuild app --minify",
+       "esbuild ssr",
        "phx.digest"
      ]
    ]
  end
As configured, this will place the generated ssr.js bundle into the priv directory. Since it's generated code, add it to your .gitignore file.
  # .gitignore

+ /priv/ssr.js

  
    
  
  Configuring your app for server-rendering


Now that you have a Node.js module capable of server-rendering your pages, youll need to tell the Inertia.js Phoenix library to perform SSR.
First, add the Inertia.SSR module to your application's supervision tree.
  # lib/my_app/application.ex

  defmodule MyApp.Application do
    use Application

    @impl true
    def start(_type, _args) do
      children = [
        MyAppWeb.Telemetry,
        MyApp.Repo,
        {DNSCluster, query: Application.get_env(:MyApp, :dns_cluster_query) || :ignore},
        {Phoenix.PubSub, name: MyApp.PubSub},
        # Start the Finch HTTP client for sending emails
        {Finch, name: MyApp.Finch},
        # Start a worker by calling: MyApp.Worker.start_link(arg)
        # {MyApp.Worker, arg},

+       # Start the SSR process pool
+       # You must specify a `path` option to locate the directory where the `ssr.js` file lives.
+       {Inertia.SSR, path: Path.join([Application.app_dir(:my_app), "priv"])},

        # Start to serve requests, typically the last entry
        MyAppWeb.Endpoint,
      ]
Then, update your config to enable SSR (if you'd like to enable it globally).
  # config/config.exs

  config :inertia,
    # The Phoenix Endpoint module for your application. This is used for building
    # asset URLs to compute a unique version hash to track when something has
    # changed (and a reload is required on the frontend).
    endpoint: MyAppWeb.Endpoint,

    # An optional list of static file paths to track for changes. You'll generally
    # want to include any JavaScript assets that may require a page refresh when
    # modified.
    static_paths: ["/assets/app.js"],

    # The default version string to use (if you decide not to track any static
    # assets using the `static_paths` config). Defaults to "1".
    default_version: "1",

    # Enable server-side rendering for page responses (requires some additional setup,
    # see instructions below). Defaults to `false`.
-   ssr: false
+   ssr: true

    # Whether to raise an exception when server-side rendering fails (only applies
    # when SSR is enabled). Defaults to `true`.
    #
    # Recommended: enable in non-production environments and disable in production,
    # so that SSR failures will not cause 500 errors (but instead will fallback to
    # CSR).
    raise_on_ssr_failure: config_env() != :prod

  
    
  
  Installing Node.js in your production


You need to have Node.js installed in your production server environment, so that we can call the SSR script when serving pages. These steps assume you are deploying your application using a Dockerfile and releases.
If you haven't installed node into your runner image, add the following command to your Dockerfile (after the FROM ${RUNNER_IMAGE} step).
  FROM ${RUNNER_IMAGE}

  # install curl (and a few other packages)
  RUN apt-get update -y && \
-     apt-get install -y libstdc++6 openssl libncurses5 locales ca-certificates && \
+     apt-get install -y libstdc++6 openssl curl libncurses5 locales ca-certificates && \
      apt-get clean && rm -f /var/lib/apt/lists/*_*

  # install Node.js
+ RUN curl -fsSL https://deb.nodesource.com/setup_x.x | bash - && \
+    apt-get update && \
+    apt-get install -y nodejs

  # ...

  ENV MIX_ENV="prod"

  # ensure node is running in production mode
+ ENV NODE_ENV="production"
[!IMPORTANT]
Be sure to set NODE_ENV=production, so that the SSR script is cached in memory. Otherwise, your page rendering times will be very slow!


  
    
  
  Client side hydration


Follow the instructions from the Inertia.js docs for updating your client-side code to hydrate the pre-rendered HTML coming from the server.
Using our example React script from above, the adaptation looks like this:
  // assets/js/app.jsx

  import React from "react";
  import { createInertiaApp } from "@inertiajs/react";
- import { createRoot } from "react-dom/client";
+ import { hydrateRoot } from "react-dom/client";

  createInertiaApp({
    resolve: async (name) => {
      return await import(`./pages/${name}.jsx`);
    },
    setup({ App, el, props }) {
-     createRoot(el).render(<App {...props} />);
+     hydrateRoot(el, <App {...props} />);
    },
  });

Maintained by the team at SavvyCal

  2.5.1



  
    
  
  Fixed


	Treat Igniter as an optional dependency in the mix inertia.install task definition. Previously, compilation would fail if Igniter was not installed.


  
    
  
  2.5.0



  
    
  
  Added


	Add inertia_deep_merge/1 function to support deep merging props on the client side (https://github.com/inertiajs/inertia/pull/2069) (#54).
	Add Igniter installer task (#51).


  
    
  
  2.4.0



  
    
  
  Added


	Add inertia_errors/1 test helper to fetch Inertia errors (#43).


  
    
  
  2.3.0



  
    
  
  Added


	Add a force_inertia_redirect plug function to instruct the client-side to always perform a full browser redirect when a redirect response is sent (#35).


  
    
  
  Changed


	Define an Inertia.Errors protocol with default implementations for Ecto.Changeset and Map.


  
    
  
  2.2.0



  
    
  
  Added


	Add preserve_case helper to prevent auto-camelization of specified prop keys.
	Add Inertia.Controller.inertia_response?/1 helper to determine if a response is Inertia-rendered.


  
    
  
  Fixed


	Ensure prop keys are compared in the proper casing (for partial reloads) when camelize_props is enabled.
	Fix prop resolution for deferred/optional props.


  
    
  
  2.1.0



  
    
  
  Fixed


	Include new Inertia v2 attributes in the initial page object (mergeProps, deferredProps, encryptHistory, clearHistory).
	Mark internal component functions in Inertia.HTML as private.


  
    
  
  2.0.0



  
    
  
  Added


	Add support new Inertia.js v2.0.0.	Add encrypt_history function to instruct the client-side to encrypt the history entry.
	Add clear_history function to instruct the client-side to clear history.
	Add inertia_optional function, to replace the now-deprecated inertia_lazy function.
	Add inertia_merge function to instruct the client-side to merge the prop value with existing data.
	Add inertia_defer function to instruct the client-side to fetch the prop value immediately after initial page load.


	Add helpers for testing Inertia-based controller responses via the Inertia.Testing module.
	Add a camelize_props global config option and a camelize_props function (to use on a per-request basis) to automatically convert prop keys from snake case to camel case.
	Accept an ssr option on the render_inertia function.


  
    
  
  Changed


	Update Phoenix LiveView to v1.0.
	The errors serializer (for Ecto.Changeset structs) has been adjusted to better align with the behavior in the Laravel adapter in cases when there are multiple validation errors for a single field.

Old behavior for errors serializer
Previously, the serializer would include each error under a separate key, with a [0] index suffix, like this:
{
  "name[0]": "is too long",
  "name[1]": "is not real"
}
While this retains maximal information about all the errors for a field, in practice it's difficult to target the right error records for display in the UI.
New behavior for errors serializer
Now, the serializer simply takes the first error message and returns it under the field name, without any added suffix:
{
  "name": "is too long"
}

  
    
  
  Fixed


	Allow for external redirects from PUT / PATCH / DELETE requests (#22)
	Camelize prop names inside lists (e.g. assign_prop(:items, [%{item_name: "..."}])).


  
    
  
  Deprecated


	The inertia_lazy/1 function has been deprecated in favor of inertia_optional/1


  
    
  
  0.10.0



  
    
  
  Bug Fixes


	Remove unsupported dot-notation in partial requests (related to inertiajs/inertia-laravel#641)


  
    
  
  0.9.0



  
    
  
  Bug Fixes


	Fix improper elimination of nested props when using only partials


  
    
  
  0.8.0



  
    
  
  Features


	Support unicode props (by using the binary flag on Node function calls)


  
    
  
  0.7.0



  
    
  
  Bug Fixes


	Fix exception when assigning structs as prop values (like DateTime)


  
    
  
  0.6.0



  
    
  
  Bug Fixes


	Prevent overly greedy empty object elimination (#14)


  
    
  
  0.5.0


	Assign errors via an assign_errors helper (#10)
	Preserve assigned errors across redirects (#10)
	Set up external redirects properly for Inertia requests (#11)
	Pass CSRF tokens via cookies (#12) 
	Forward flash contents across forced refreshes (#13)
	Automatically pass Phoenix flash data via the flash prop


  
    
  
  0.4.0


	Support for partial reloads (#6)
	Support lazy data evaluation (#7)


  
    
  
  0.3.0


	Add raise_on_ssr_failure configuration


  
    
  
  0.2.0


	Add SSR support
	Add <.inertia_head> component for rendering head elements provided by SSR


  
    
  
  0.1.0


	Initial release



  

    
License
    

MIT License
Copyright (c) 2024-current SavvyCal, Inc.
Permission is hereby granted, free of charge, to any person obtaining a copy
of this software and associated documentation files (the "Software"), to deal
in the Software without restriction, including without limitation the rights
to use, copy, modify, merge, publish, distribute, sublicense, and/or sell
copies of the Software, and to permit persons to whom the Software is
furnished to do so, subject to the following conditions:
The above copyright notice and this permission notice shall be included in all
copies or substantial portions of the Software.
THE SOFTWARE IS PROVIDED "AS IS", WITHOUT WARRANTY OF ANY KIND, EXPRESS OR
IMPLIED, INCLUDING BUT NOT LIMITED TO THE WARRANTIES OF MERCHANTABILITY,
FITNESS FOR A PARTICULAR PURPOSE AND NONINFRINGEMENT. IN NO EVENT SHALL THE
AUTHORS OR COPYRIGHT HOLDERS BE LIABLE FOR ANY CLAIM, DAMAGES OR OTHER
LIABILITY, WHETHER IN AN ACTION OF CONTRACT, TORT OR OTHERWISE, ARISING FROM,
OUT OF OR IN CONNECTION WITH THE SOFTWARE OR THE USE OR OTHER DEALINGS IN THE
SOFTWARE.


  

    
Inertia 
    



      
The Inertia.js adapter for Elixir/Phoenix applications.

      




  

    
Inertia.Controller 
    



      
Controller functions for rendering Inertia.js responses.

      


      
        Summary


  
    Types
  


    
      
        
  
    
    Inertia.Errors - Inertia v2.5.1
    
    

    


  
  

    
Inertia.Errors protocol
    



      
Converts a value to Inertia.js-compatible validation
errors.
This protocol allows you to transform various error structures into the format
expected by Inertia.js for client-side validation. The protocol converts error
structures into a flat map where keys represent field paths and values are error
messages.

  
    
  
  Built-in Implementations


This library includes default implementations for:
	Ecto.Changeset - Converts changeset errors to Inertia-compatible format
	Map - Validates and passes through properly formatted error maps


  
    
  
  Usage with Ecto.Changeset


# In your controller action
def create(conn, %{"post" => post_params}) do
  case Posts.create(post_params)
    # Handle successful validation
    {:ok, post} ->
      redirect(conn, to: ~p"/posts/#{post}")

    # Convert changeset errors and share them with Inertia
    {:error, changeset} ->
      conn
      |> assign_errors(changeset)
      |> redirect(conn, to: ~p"/posts/new")
  end
end
The assign_errors/2 function is a convenience helper provided by Inertia.Controller
that internally uses Inertia.Errors.to_errors/1 to serialize the changeset errors
and share them with the Inertia page under the errors key.

  
    
  
  Custom Error Formatting


You can provide a custom message formatting function as the second argument to
to_errors/2:
Inertia.Errors.to_errors(changeset, fn {msg, opts} ->
  # Custom error message formatting logic
  Gettext.dgettext(MyApp.Gettext, "errors", msg, opts)
end)

  
    
  
  Implementing for Custom Types


You can implement this protocol for your own error types:
defimpl Inertia.Errors, for: MyApp.ValidationError do
  def to_errors(validation_error) do
    # Convert your custom error structure to a map of field paths to error
    # messages
    %{
      "field_name" => validation_error.message,
      "nested.field" => "Another error message"
    }
  end

  def to_errors(validation_error, _msg_func) do
    # Custom implementation with message function
    to_errors(validation_error)
  end
end

      


      
        Summary


  
    Types
  


    
      
        
  
    
    Inertia.HTML - Inertia v2.5.1
    
    

    


  
  

    
Inertia.HTML 
    



      
HTML components for Inertia views.

      


      
        Summary


  
    Functions
  


    
      
        
  
    
    Inertia.Plug - Inertia v2.5.1
    
    

    


  
  

    
Inertia.Plug 
    



      
The plug module for detecting Inertia.js requests.

      


      
        Summary


  
    Functions
  


    
      
        
  
    
    Inertia.SSR - Inertia v2.5.1
    
    

    


  
  

    
Inertia.SSR 
    



      
A supervisor that provides SSR support for Inertia views. This module is
responsible for starting a pool of Node.js processes that can run the SSR
rendering function for your application.

      


      
        Summary


  
    Functions
  


    
      
        
  
    
    Inertia.Testing - Inertia v2.5.1
    
    

    


  
  

    
Inertia.Testing 
    



      
Helpers for testing Inertia responses.

      


      
        Summary


  
    Functions
  


    
      
        
  
    
    Inertia.SSR.RenderError - Inertia v2.5.1
    
    

    


  
  

    
Inertia.SSR.RenderError exception
    






  

  
    
    mix inertia.install - Inertia v2.5.1
    
    

    


  
  

    
mix inertia.install 
    



      
Installs and configures the Inertia.js adapter in a Phoenix application.
This installer:
	Updates controller and HTML components to import Inertia functions
	Adds the Inertia plug to the browser pipeline
	Adds basic configuration in config.exs
	Updates esbuild and configures code splitting
	Set up the client-side integration packages


  
    
  
  Usage


mix inertia.install


  
    
  
  Options


--client-framework FRAMEWORK  Framework to use for the client-side integration
                              (react, vue, or svelte). Default is react.
--camelize-props              Enable camelCase for props
--history-encrypt             Enable history encryption
--typescript                  Enable TypeScript
--yes                         Don't prompt for confirmations

